
RAPID PROTOTYPING IN PVS∗
César A. Muñoz†

ABSTRACT
PVSio is a conservative extension to the PVS prelude library that provides basic input/output capabilities to the PVS ground evaluator. It supports rapid prototyping
in PVS by enhancing the specification language with built-in constructs for string
manipulation, floating point arithmetic, and input/output operations.

1

INTRODUCTION

PVS [6] is a verification system based on a typed classical higher-order logic enriched with
predicate subtyping and dependent records [7]. The system is widely known by its expressive
specification language and its powerful theorem prover. The ground evaluator, which was
originally announced as an experimental feature in PVS Release 2.3 [10], is a less known
feature of PVS that allows for the animation of functional specifications: it extracts efficient
Common Lisp code for a large set of PVS constructs [8].
The ground evaluator is a fundamental mechanism for rapid prototyping in PVS. However, it does not provide typical features of programming languages such as input/output
functionality or floating-point arithmetic. For instance, a ground expression like sqrt(2),
where sqrt is defined as in the NASA reals library [1], is not handled by the ground evaluator.
This lack of capabilities greatly limits the usefulness of the evaluator.
Another experimental feature of PVS, even less known than the ground evaluator, is called
semantic attachments [4]. Semantic attachments are user-defined Common Lisp functions
that the ground evaluator calls during the evaluation of PVS expressions. For instance,
given an appropriate semantic attachment, the expression sqrt(2) can be evaluated as the
Common Lisp expression (sqrt 2). Semantic attachments must be handled with care.
Indeed, the ground evaluator does not enforce type correctness of semantic attachments nor
does it check that semantic attachments respect the intended semantics of the corresponding
PVS expressions. Hence, semantic attachments may break the ground evaluator. However,
since neither the ground evaluator nor the semantic attachments are integrated into the
logical framework of PVS, the soundness of the theorem prover is not compromised.
PVSio is a prelude library extension implemented on top of semantic attachments that
relieves PVS users from all the burden and technical details of Common Lisp programming of
semantic attachments. PVSio enhances the specification language with built-in constructs
for string manipulation, floating-point arithmetic, and input/output operations. From a
logical point of view, PVSio is a conservative extension to the PVS prelude library.

GROUND EVALUATOR

The PVS ground evaluator consists of a PVS to Common Lisp translator, an interactive readeval-print interface, and a proof rule. The translation itself and the fragment of PVS that
is suitable for ground evaluation are described in [10]. Roughly speaking, the unexecutable
constructs of PVS are: uninterpreted symbols, non-bounded quantifications, and higherorder relations. However, evaluations are non-strict, that is, unexecutable terms may still
occur in executable expressions as long as they are not required in the overall evaluation.
The ground evaluator assumes that Type Correctness Conditions (TCCs) associated with
executable expressions have been properly discharged. Unsound evaluations may result from
unproven TCCs. For rapid prototyping, it is a working conjecture that ground evaluations
are sound with respect to valid TCCs.
The interactive interface of the PVS ground evaluator is invoked with the Emacs command M-x pvs-ground-evaluator. It consists of a read-eval-print loop where user inputs
are prompted by <GndEval>. Control commands such as quit, help, quiet, etc., may be
typed after the prompt1 . PVS expressions are supplied to the evaluator enclosed in double
quotes ("). This notation is unfortunate since it requires PVS strings to be escaped when
they are provided to the interactive evaluator interface. For instance, the PVS string "Hello
World" has to be written \"Hello World\" within the read-eval-print loop.
The ground evaluator is also available through the proof rule EVAL in the theorem prover.
The proof command (EVAL e) translates the ground expression e into Common Lisp, evaluates it, and displays the result. In contrast to expressions in the interactive environment, e is
not enclosed in double quotes and, consequently, PVS strings may be used without escaping
them. From a logical point of view, EVAL behaves like SKIP when it terminates. Henceforth,
for presentation purposes, we will always refer to the interactive interface of the PVS ground
evaluator. However, both interfaces provide similar functionality.
Assume the following definition of sqrt newton from the theory sqrt approx of the PVS
NASA Langley reals library: [1]
sqrt_newton(a:nonneg_real,n:nat) : RECURSIVE posreal =
IF n = 0 THEN a+1
ELSE (1/2) * (sqrt_newton(a,n-1) + (a / sqrt_newton(a,n-1)))
ENDIF
MEASURE (n+1)
It has been formally verified that sqrt newton satisfies among others the following properties:
√
1. a < sqrt newton(a, n), and
2. sqrt newton(a, n + 1) < sqrt newton(a, n).
It can also be shown that when n goes to infinity, sqrt newton(a, n) converges to
function sqrt newton can be animated in PVS:2
1

√

a. The

See [10] for a description of the commands available at the interactive interface of the PVS ground
evaluator.
2
For readability, some messages have been supressed from the output.

<GndEval> "sqrt_newton(2,1)"
==>
11/6
<GndEval> "sqrt_newton(2,2)"
==>
193/132
<GndEval> "sqrt_newton(2,3)"
==>
72097/50952
√
We manually check that 2 < 72097/50952 < 193/132 < 11/6. We soon realize that we
cannot go too far with the evaluations:
<GndEval> "sqrt_newton(2,8)"
==>
604540277611356030096574510534826474174613505606725073236811480078532
817791953637081292805551863485709675189297381081965179971672243083538
463540494545438709624073217/
427474529799387823675525584913729551524427282542777624939553930048017
831989756724882869085917429624670788187662034448252729167799126170270
230544308203049155944825088
3

OVERVIEW TO PVSIO

For rapid prototyping we clearly need a better mechanism for exchanging information between the evaluation environment and PVS. At a minimum, one such mechanism should
allow for basic input/output capabilities such as reading and pretty printing. Semantic attachments [4] provide the low level machinery needed for implementing this functionality.
However, they are difficult to use, require Common Lisp knowledge and expertise, and are
error prune.
PVSio is a PVS prelude library extension built on top of semantic attachments. It brings,
among other things, basic input/output capabilities to the PVS ground evaluator without all
the burden of semantic attachments and Common Lisp coding. For instance, after loading
the PVSio prelude library3 , we can evaluate
<GndEval> "println(sqrt_newton(2,8))"
1.4142135
==>
TRUE
If we assume the following PVS definition:
3

This is done with the Emacs command M-x load-prelude-library PVSio.

3

sqrt_io : void =
LET a = query_real("Enter a positive real number:") IN
assert(a >= 0,"Input Error") &
println("The approx. sqrt of "+a+" is: "+sqrt_newton(a,8))
we can also evaluate
<GndEval> "sqrt_io"
Enter a positive real number:
2
The approx. sqrt of 2 is: 1.4142135
==>
TRUE
<GndEval> "sqrt_io"
Enter a positive real number:
1.5
The approx. sqrt of 1.5 is: 1.2247449
==>
TRUE
<GndEval> "sqrt_io"
Enter a positive real number:
-10
Assert Failure: Input Error
==>
FALSE
Note that the values returned by println and sqrt io are Boolean. This is because
println and sqrt io are defined of type void, which is an alias to bool in PVSio. Hence,
sequential statements can be separated with the symbol &, which is a logical and in PVS.
The function sqrt io uses functions query real and assert. The former allows for
animation of expressions containing non-bounded variables that become dynamically bound
to user inputs. PVSio does not check for consistency of user-provided input data. To
partially overcome this potential source of problems, PVSio provides assert(b,s), where b
is a Boolean expression and s is a string. If b evaluates to FALSE, the message "Assertion
Failure: s" is printed. From a logical point of view, the whole expression is equivalent to
b.
We emphasize that for the PVS theorem prover there is nothing special about PVSio
functions. For instance, the proposition ∀(s) : query real(s) = query real(s) is trivially proved in PVS. However, for the PVS ground evaluator, the value of the expression
query real(s) = query real(s), for a particular value of s, depends on the user inputs:
<GndEval> "query_real(s) = query_real(s)"
s
1
4

s
2
==>
FALSE
The example above shows that, as in the case of semantic attachments, PVSio functions
may have side-effects in the evaluation environment. Hence, we may get unexpected results. However, these effects do no permeate to the PVS logical framework. In other words,
query real(s) = query real(s) may evaluate to FALSE in the ground evaluator, but this
cannot be used to prove FALSE in the theorem prover.
Input/output functionality is not the only limitation
√ of the ground evaluator. Assume
that we want to evaluate the ground expression sin( 2). Similar to sqrt newton, we may
define an executable function sin approx that converges to sine at infinity, for example
using Taylor’s series. In this
√ case, we could animate sin approx(sqrt newton(2)) to get a
good approximation of sin( 2). Such approximations of real and trigonometric functions are
already defined in theories sqrt approx and trig approx of the NASA reals and trig libraries
[1]. Functions in those theories are used in [2] to automate reasoning about non-computable
functions such as squared root and trigonometric functions. However, they are not really
suitable for rapid prototyping. Even the most efficient PVS definitions of sqrt newton and
sin approx are not as efficient as native Common Lisp functions sqrt and sin. For this
reason, PVSio enhances the ground evaluator with floating point arithmetic via semantic
attachments:
<GndEval> "print(SIN(SQRT(2)))"
0.98776597
==>
TRUE
Floating point operators are fully capitalized in PVSio to emphasize that they are not
the real functions. It is well-known that floating-point arithmetic may contradict standard
mathematical results:
<GndEval> "SQ(SIN(2))+SQ(COS(2)) = 1"
==>
FALSE
4

PVSIO LIBRARY

PVSio4 is implemented as a prelude library extension that profit from the library support
provided in PVS 3.1 [11]. It consists of the following theories:
• stdlang: Language definitions.
• stdstr: String operations.
• stdio: Input/output operations.
4

The PVSio package is freely available at http://research.nianet.org/~munoz/PVSio.

Table 1: Theory stdlang
Name
void
skip
exit
assert(b:bool,mssg:string)
assert(b:bool,s:void)
assert(b:bool)
fail(s:void)
try(s1:void,s2:void)
catch(s:void)
seq[T:TYPE](s:void,t:T)

Type
TYPE
void
void
void
void
void
void
void
void
T

Value Description
bool
Type of procedures and statements
TRUE
Empty statement
FALSE Exit statement
If NOT b then displays mssg & exit. Otherwise, skip
If NOT b then s & exit. Otherwise, skip
assert(b,skip)
s & exit
s1 OR s2
s1 OR skip
Performs s, returns t

• stdmath: Floating point arithmetic.
• stdpvs: Basic reflection.
• stdindent: Definition of indentations.
• stdtokenizer: Definition of tokenizers.
Since PVSio is a prelude extension, the user does not need to import any of these
theories. In a fresh context, the library PVSio is loaded with the Emacs command M-x
load-prelude-library PVSio.
The rest of this section describes PVSio theories in some detail. However, note that
PVSio is still under development. New theories and functions are expected to be added in
the future.
4.1

Theory stdlang: Language definitions

Table 1 summarizes the definitions in theory stdlang.5 Type void intends to serve as the
type of procedures and statements, as opposed to functions and expressions. Constants
skip and exit, both of them of type void, represent the empty and the exit statements,
respectively. Since void is equivalent to bool, the symbol & (logical and) serves as separator
of sequential statements. Furthermore, quantifiers FORALL and EXISTS, over a finite domain,
serve as iterators. PVSio does not provide assignments as statements.
Theory stdlang also defines control structures such as assert, try, fail, catch, and
seq. As we saw in Section 3, assert provides a mechanism for simple runtime verification
of Boolean conditions. Operations fail, try, and catch complement the functionality
of assert with basic handling of exceptional cases. Finally, seq allows for sequencing of
statements, possibly having side-effects, and PVS expressions. Type T is usually inferred by
PVS, but in some cases it has to be provided by the user.
5

For technical reasons, assert(b:bool,mssg:string) is defined in theory stdio, and seq is defined in
theory stdproc.

Table 2: Theory stdstr
Name
empty
space
newline
tab
quote
spaces(n:nat)
upcase(s:string)
downcase(s:string)
capitalize(s:string)
substr(s:string,i,j:nat)
find(s1,s2:string)
fill(n,s:string)
str2real(s:string)
str2int(s:string)
str2bool(s1,s2:string)
number?(s:string)
int?(s:string)
concat(s1,s2:string)
#(s:string)
tostr(e)
e1 + e2

Type
string
string
string
string
string
string
string
string
string
string
int
string
rat
int
bool
bool
bool
bool
rat
string
string

Description
Empty string
Space
New line
Tab
Double quote
n spaces
Uppercase of s
Lowercase of s
Capitalization of s
Substring of s from i to j-1
Index of leftmost occurrence of s2 in s1 (or -1)
n times string s
Rational denoted by s (s is in decimal notation)
Integer denoted by s
downcase(s1) = downcase(s2)
Tests if string s represents a numerical value
Tests if string s represents an integer value
Concatenation of s1 and s2
str2real(s)
String representation of e
concat(e1 ,e2 )

Note that most functions in stdlang are limited to statements, i.e., expressions of type
void. PVSio does not waive PVS type checking. For example, if the domain of a function is
real, it cannot return exit. Furthermore, stdlang operations do not extend to Common
Lisp exceptions. Currently, PVSio does not provide a mechanism for handling Common Lisp
exceptions.
4.2

Theory stdstr: String operations

Table 2 summarizes the definitions in theory stdstr. Note that PVS does not support decimal notation of real numbers. PVSio alleviates this limitation with the function str2real
(also denoted with the symbol #). It takes the string representation of a number in decimal notation and returns its value as a rational number. For instance, the string #("1.5")
evaluates to the PVS number 3/2.
The symbols tostr and + are overloaded for basic values, that is, e, e1 , e2 may be of type
real, bool, or string. Since tostr is a PVS conversion, basic values are automatically
coerced to string values when necessary. Indeed, the symbol + behaves pretty much as +
in Java. The practice of having conversions from user-defined types to string is strongly
advised to fully exploit the functionality provided by stdstr.

Examples:
<GndEval> "#(\"1.5\") + #(\"2.3\") - 1"
==>
14/5
<GndEval> "1+tab+12+tab+123+tab+1234+tab+12345+tab"
==>
"1
12
123
1234
12345
"
<GndEval> "\"Boolean: \"+(1=1)"
==>
"Boolean: TRUE"
4.3

Theory stdio: Input/Output operations

Table 3 summarizes the definitions in theory stdio. This theory implements classical I/O
functionality as reading from standard input, printing to standard output, and reading and
writing to character streams. As said before, PVSio does not check for consistency of input
data. Hence, the user is responsible for providing the appropriate kind of data, otherwise
the ground evaluator will break.
Input operations are based on Common Lisp reading conventions, which are fairly complex (see [9], Section Input/Output). For the sake of simplicity, we assume the following
lexical rules:
• Integer: Sequence of numerical characters.
• Real: Sequence of numerical characters and dot symbol (.).
• Word: Sequence of alpha-numerical characters and symbols in !?@#$%^& */+-=|{}[]<>
separated by space, newline, or tab.
• Line: Sequence of symbols terminated by newline.
Symbols in (),.:;\~‘’" have special meaning in Common Lisp and they are not handled
correctly by PVSio.
Character streams are defined by an uninterpreted type Stream. Input and output
streams, IStream and OStream, respectively, are defined as subtypes of Stream. A stream
is typically associated with a file. In this case they are called file streams. PVSio also allows
string streams, that is, streams associated with strings. Except for the functions flength,
that only applies to file streams, and the function sget, that only applies to output string
streams, both kinds of streams are handled indistinctly by PVSio.
An input stream can be created in two modes: input (which is the default) and string.
For instance, open in(input,s) opens an input file stream associated to the file named
s. PVSio assumes that the file does exist. The existence of such a file can be tested with
fexists(s). The operation open in(string,s) opens an input string stream that reads
information from string s.

8

Table 3: Theory stdio
Name

Type
Description
Standard Input
query int(s:string)
int
Prompts s, reads and returns an integer
query real(s:string)
rat
Prompts s, reads a decimal and returns it as rat
query word(s:string)
string
Prompts s, reads and returns a string word
query line(s:string)
string
Prompts s, reads and returns a string line
query bool(s1,s2:string)
bool
str2bool(query word(s1),s2)
read int
int
query int(empty)
read real
rat
query real(empty)
read word
string
query word(empty)
read line
string
query line(empty)
read bool(s:string)
bool
query bool(empty,s)
Standard Output
print(s:string)
void
Prints string s
println(s:string)
void
Prints string s+newline
I/O Character Streams
Stream
TYPE+
Uninterpreted type of character streams
IStream
⊆ Stream Uninterpreted type of input character streams
OStream
⊆ Stream Uninterpreted type of output character streams
Mode
TYPE
{input,new,append,overwrite,string}
stdin
IStream
Standard input stream
stdout
OStream
Standard output stream
open in(m:Mode,s:string)
IStream
Opens an input stream
open in(s:string)
IStream
open in(input,s)
open out(m:Mode,s:string)
OStream
Opens an output stream
open out(s:string)
OStream
open out(new,s)
sget(f:OStream)
string
String from output string stream f
close(f:Stream)
void
Closes f
eof ?(f:Stream)
bool
Tests the end of stream f
flength(f:Stream)
nat
Length of file stream f
fexists(s:string)
bool
Tests if there exists a file with name s
Input from Character Streams
scan int(f:IStream)
int
Reads an integer from f and returns it
scan real(f:IStream)
rat
Reads a decimal from f and returns it as rat
scan word(f:IStream)
string
Reads a string word from f and returns it
scan line(f:IStream)
string
Reads a string line from f and returns it
scan bool(f:IStream,s:string) bool
str2bool(scan word(f),answer)
Output to Character Streams
print(f:OStream,s:string)
void
Writes string s to f
println(f:OStream,s:string)
void
Writes string s+newline to f
echo(f:OStream,s:string)
void
Writes string s to f and to stdout
echoln(f:OStream,s:string)
void
Writes string s+newline to f and to stdout
9

An output stream can be created in several modes: new (which is the default), append,
overwrite, and string. All the modes in {new,append,overwrite} open output file
streams. They differ in the action they perform should a file with the same name already exist. The mode new supersedes any file with the same name, the mode append writes at the end
of the old file, and the mode overwrite writes at the beginning of the old file overriding the
content of the file as output operations are performed. The operation open out(string,s)
opens an output string stream (the value of s is ignored). If f is an output string stream,
the content of the string that is being written is obtained with sget(f ).
Examples:
<GndEval> "read_line"
Hello World
==>
"Hello World"
<GndEval> "read_int + read_real"
10
10.5
==>
41/2
<GndEval> "LET i=query_int(\"Give me a nat:\") IN print(1+spaces(i)+1)"
Give me a nat:
4
1
1
==>
TRUE
<GndEval> "println(\"Hello World\")"
Hello World
==>
TRUE
<GndEval> "print(1=0) & println(123) & println(1/2)"
FALSE123
0.500000
==>
TRUE
4.4

Theory stdmath: Floating point arithmetic

Table 4 summarizes the definitions in theory stdmath. Names are fully capitalized to emphasize that they are floating point operations (not the real ones). The constant NaN is
returned by functions in stdmath when the result is not well-defined, e.g., SQRT(-1) = NaN.
However, since NaN is not a ground expression, it cannot be evaluated by the PVS ground
evaluator. NaN may still be part of an executable expression when its value is not required
in the overall evaluation.
10

Table 4: Theory stdmath
Name
NaN
SQ(x:real)
SQRT(x:real)
SIN(x:real)
COS(x:real)
TAN(x:real)
ASIN(x:real)
ACOS(x:real)
ATAN(y,x:real)
PI
NPI(n:int)
RANDOM
NRANDOM(n:posnat)
BRANDOM
RAD2DEG(x:real)
DEG2RAD(x:real)
TRUNC(x:real)
ROUND(x:real)
DIV(n:nat,m:posnat)
MOD(n:nat,m:posnat)

Type
number
nnreal
nnreal
real
real
real
real
real
real
real
real
real
nat
bool
real
real
int
int
nat
nat

Description
Uninterpreted number
x2
√
≈ x
≈ sin x
≈ cos x
≈ tan x
≈ sin−1 x
≈ cos−1 x
≈ tan−1 x/y
≈π
≈ nπ
Pseudo-random number in the real interval [0, 1)
Pseudo random number in the natural interval [0, n)
Pseudo random Boolean
≈ 180x/π
≈ xπ/180
Truncation of x
Rounding of x
Euclidean division of n by m
Remainder of Euclidean division of n by m

11

Table 5: Theory stdpvs
Name
Type
Description
str2PVS[T](s:string)
T
Translates a string to a PVS expression
PVS2str(type:string,t:T) string Translates a PVS expression to a string

4.5

Theory stdpvs: Reflection

PVSio provides very basic capabilities for reflection through the theory stdpvs (Table 5).
Strings are translated back and forth to PVS via str2PVS and PVS2str. For example,
assuming the following definitions in PVS
REC : TYPE = [# x : real, y: real #]
rec : REC = (# x := 5, y:= 6 #)
we get
<GndEval> "str2PVS[REC](\"rec\")"
==>
(# x := 5, y := 6 #)
<GndEval> "PVS2str(\"REC\",rec)"
==>
"(# x := 5, y := 6 #)"
Note that the type of the expression is given as a theory parameter in str2PVS, whereas it
is given as a string in PVS2str. Full reflection in PVSio is limited by the fact that the PVS
ground evaluator ignores all type information from PVS expressions when translating them
to Common Lisp. Therefore, there is no a simple way to reconstruct a PVS type from the
Common Lisp translation of a PVS ground expression.
4.6

Theory stdindent: Indentations

Theory stdindent (Table 6) defines indentations. An indentation is composed of a built-in
stack of integers, a predefined number of spaces, and a prefix. Indentations are suitable for
printing block-formatted text, such as source code, or linearization of tree-structures. The
use of indentation is best illustrated with some examples.6
Assume the definitions
chello : void =
LET i = create_indent(2) IN
prindentln(i,"void main() {") & open_block(i) &
prindentln(i,"while(1) {") & open_block(i,3) &
prindentln(i,"println("+quote+"Hello World"+quote+");") & close_block(i) &
6

These examples are available as part of the standard distribution of PVSio.

Table 6: Theory stdindent
Name
Indent
create indent(n:nat,s:string)
create indent(n:nat)
open block(i:Indent,n:nat)
open block(i:Indent)
close block(i:Indent)
get indent(i:Indent)
set indent(i:Indent,n:nat)
get prefix(i:Indent)
set prefix(i:Indent,s:string)
indent(i:Indent,s:string)
prindent(i:Indent,s:string)
prindentln(i:Indent,s:string)
prindent(f:OStream,i:Indent,s:string)
prindentln(f:OStream,i:Indent,s:string)

Type
TYPE+
Indent
Indent
void
void
void
nat
void
string
void
string
void
void
void
void

Description
Uninterpreted type for indentations
Indentation of n spaces and prefix s
Indentation of n spaces without prefix
Opens block of n spaces
Opens block with default spaces
Closes block
Default number of spaces
Sets the default number of spaces to n
Default prefix
Sets the default prefix to s
Indented s
Prints indented s
Prints indented s + newline
Writes indented s to f
Writes indented s + newline to f

prindentln(i,"}") & close_block(i) &
prindentln(i,"}")
stair : void =
LET i = create_indent(3,"=>") IN
(FORALL(n:subrange(1,3)): prindentln(i,n) & open_block(i)) &
set_indent(i,1) &
(FORALL(n:subrange(4,6)): prindentln(i,n) & open_block(i)) &
set_prefix(i,"<=") &
(FORALL(n:subrange(0,6)): prindentln(i,n) & close_block(i))
Then,
<GndEval> "chello"
void main() {
while(1) {
println("Hello World");
}
}
==>
TRUE
<GndEval> "stair"
=>1
=>
2

=>
=>
=>
=>
<=
<=
<=
<=
<=
<=
5
<=6
==>
TRUE
4.7

3
4
5
6
0
1
2
3
4

Theory stdtokenizer: Tokenizers

Theory stdtokenizer (Tables 7 and 8) is the most elaborate theory of PVSio. By itself,
stdtokenizer is a good example of what can be achieved with the PVSio package. Indeed,
the entire theory is written as a PVSio application.
A tokenizer is a list of tokens, i.e., string words, with basic parsing capabilities. The
structure of the type Tokenizer is not relevant here. As structured programming mandates,
users should access the internal structure only through the public interface. Tokenizers can
be created from files through file2tokenizer(s), where s is the name of the file, or from
strings through str2tokenizer(s), where s is the string containing the tokens. Once a
tokenizer is created, go next and several accept-functions process the tokens one by one.
Errors are indicated by error?(t) and error(t). User-defined errors are allowed through
set error(t,code), where negative codes are reserved for this purpose. Standard error codes
are listed in Table 9.
Error messages are available through Messenger objects, which are functions from int
to string. Standard error messages for tokenizer t are retrieved with function std mssg(t).
The statement print error(t) prints an error message for t from the standard messenger.
Furthermore, standard errors messages can be overridden by using mssg * (code,s). If mssg
is a messenger, print error(t,mssg) prints an error message for t from mssg.
We illustrate the use of stdtokenizer with a simple example.7 Assume that we want to
parse a list of natural numbers given as a string "< n1 & ... & nk >", where 1 ≤ ni ≤ 5
for 0 ≤ i ≤ k (i = 0 being the empty list). The PVS theory parse in Figure 1 implements
that informal specification (lines have been numbered for the discussion).
• Line 4 defines the test that numbers in the list should satisfy, i.e., 1 ≤ ni ≤ 5.
• Lines 5 to 10 define error messages to be printed in case of a syntax error. We have
decided to override standard messages for EndOfTokenizer and ExpectingTestWord,
and to add new codes −1 and −2 for special error messages.
• The parser itself is implemented in lines 11 to 30 as a tail recursive function. The
parameters of that function are the current tokanizer t and the accumulated list l.
7

The example is available as part of the standard distribution of PVSio.

Table 7: Theory stdtokenizer: Declarations
Name
Tokenizer
file2tokenizer(s:string)
str2tokenizer(s:string)
tostr(t:Tokenizer)
error(t:Tokenizer)
error?(t:Tokenizer)
set error(t:Tokenizer,code:int)
token(t:Tokenizer)
next token(t:Tokenizer)
token at(t:Tokenizer,n:nat)
val int(t:Tokenizer)
val real(t:Tokenizer)
eot?(t:Tokenizer)
line(t:Tokenizer)
length(t:Tokenizer)
pos(t:Tokenizer)
go next(t:Tokenizer)
go back(t:Tokenizer)
accept word(t:Tokenizer,test:[string->bool])
accept word(t:Tokenizer,s:string)
accept word(t:Tokenizer)
accept int(t:Tokenizer,test:[int->bool])
accept int(t:Tokenizer)
accept real(t:Tokenizer,test:[real->bool])
accept real(t:Tokenizer)
Messenger
std mssg(t:Tokenizer)
*(mssg:Messenger,(code:int,s:string))
print error(t:Tokenizer)
print error(t:Tokenizer,mssg:Messenger)

Type
TYPE
Tokenizer
Tokenizer
string
int
bool
Tokenizer
string
string
string
int
real
bool
nat
nat
upto(length(t))
Tokenizer
Tokenizer
Tokenizer
Tokenizer
Tokenizer
Tokenizer
Tokenizer
Tokenizer
Tokenizer
TYPE
Messenger
Messenger
void
void

Table 8: Theory stdtokenizer: Descriptions
Name
file2tokenizer(s)
str2tokenizer(s)
tostr(t)
error(t)
error?(t)
set error?(t,code)
token(t)
next token(t)
token at(t,n)
val int(t)
val real(t)
eot?(t)
line(t)
length(t)
pos(t)
go next(t)
go back(t)
accept word(t,test)
accept word(t,s)
accept word(t)
accept int(t,test)
accept int(t)
accept real(t,test)
accept real(t)
Messenger
std mssg(t)
mssg * (code,s)
print error(t)
print error(t,mssg)

Description
Creates a tokenizer from file named s
Creates a tokenizer from string s
String representation of tokenizer t
Error code of the tokenizer
Tests if tokenizer has found a syntax error
Sets code error to code
Current token (token already processed)
Next token (next token to process)
Token at position n
Integer value of current token or 0
Real value of current token or 0
Tests for end of tokenizer
Line number of current token
Length of tokenizer
Current position of tokenizer
Processes next token
Goes to previous processed token
Accepts word that satisfies test
Accepts word s
Accepts any non-numerical value
Accepts integer that satisfies test
Accepts any integer
Accepts real that satisfies test
Accepts any real
[int->string]
Standard messenger for t
Overrides messenger mssg with message s in code
Prints error message from std mssg
Prints error message from messenger mssg

1: parse : THEORY
2: BEGIN
3:

IMPORTING list[int]

4:

test15(n:int):bool = 1 <= n AND n <= 5

5:
6:
7:
8:
9:
10:

mssg15(t:Tokenizer):[int->string] =
std_mssg(t) *
(EndOfTokenizer,"Truncated list") *
(ExpectingTestWord,"Expecting <. Found: "+next_token(t)) *
(-1,"Expecting 1 <= n <= 5. Found: "+next_token(t)) *
(-2,"Expecting either & or >. Found: "+next_token(t))

11:
12:
13:
14:
15:
16:
17:
18:
19:
20:
21:
22:
23:
24:
25:
26:
27:
28:
29:
30:

parse15(t:Tokenizer,l:list) : RECURSIVE [Tokenizer,list] =
IF error?(t) THEN (t,null)
ELSIF eot?(t) THEN (set_error(t,EndOfTokenizer),null)
ELSIF number?(next_token(t)) THEN
LET t = accept_int(t,test15) IN
IF error?(t) THEN
(set_error(t,-1),null)
ELSIF next_token(t) = "&" THEN
parse15(accept_word(t,"&"),cons(val_int(t),l))
ELSIF next_token(t) = ">" THEN
(t,reverse(cons(val_int(t),l)))
ELSE
(set_error(t,-2),null)
ENDIF
ELSIF null?(l) AND next_token(t) = ">" THEN
(t,null)
ELSE
(set_error(t,-1),null)
ENDIF
MEASURE IF error?(t) THEN 0 ELSE (length(t)-pos(t)+1) ENDIF

31:
32:
33:
34:

parser(s:string) : list =
LET t = str2tokenizer(s) IN
LET (t,l) = parse15(accept_word(t,"<"),null) IN
seq(print_error(t,mssg15(t)),l)

35: END parse
Figure 1: Example of stdtokenizer capabilites

Table 9: Error codes for stdtokenizer
Code
NoError
FileNotFound
EndOfTokenizer
InvalidToken
ExpectingWord
ExpectingTestWord
ExpectingInt
ExpectingTestInt
ExpectingReal
ExpectingTestReal

Value
0
1
2
3
4
5
6
7
8
9
n<0

Description
No error has been found
File not found (name of the file available at next token(t))
End of tokenizer has been found
Invalid token
Expecting a word
Expecting word that satisfies test
Expecting an integer
Expecting integer that satisfies test
Expecting a real
Expecting real that satisfies test
Reserved for user-defined errors

After processing the tokens, the function returns a tokenizer and the constructed list
(which has been reversed in line 21 to return it in the right order). Line 30 gives the
measure that guarantees termination.
• Lines 31 to 34 define a simple function that tests the parser. It has as input the string
to be processed and as output the list that has been constructed. If the parser finds
a syntax error an error message is printed and the list null is returned. If instead of
reading the tokens from a string, we would like to read them from a file called s, we
simply write LET t = file2tokenizer(s) IN in line 32.
Finally, we test our parser with some examples:
<GndEval> "parser(\"< >\")"
==>
(: :)
<GndEval> "parser(\"< 5 >\")"
==>
(: 5 :)
<GndEval> "parser(\"< 2 & 4 & 3 & 1 >\")"
==>
(: 2, 4, 3, 1 :)
<GndEval> "parser(\"< 2 4 3 1 \")"
Syntax Error. Line 1: Expecting either & or >. Found: 4
==>
(: :)

<GndEval> "parser(\"< 20 >\")"
Syntax Error. Line 1: Expecting 1 <= n <= 5. Found: 20
==>
(: :)
The example above illustrates the functionality provided by theory stdtokenizer. However, the simplest way of parsing PVS expressions, such as lists and data types, is using
the standard PVS lexer and parser via str2PVS (see Section 4.5). In that case, syntax and
type-checking errors are reported directly by PVS, and, therefore, the user has no control
on the error handling mechanism.
5

CONCLUSION

Rapid prototyping is the ability to execute a system from the very early stages of its development. Formal techniques address rapid prototyping by extracting code from formal
specifications. The PVS ground evaluator, for instance, extracts very efficient Common Lisp
code from PVS specifications. If we trust the implementation of the verification system (including the theorem prover and the ground evaluator), the PVS ground evaluator produces
a correct and efficient Common Lisp implementation from the functional specification of an
algorithm,
Extraction of efficient code is not enough for rapid prototyping. Usually, formal specifications languages, such as PVS, include a large subset of executable functions that may act
as a functional programming language. However, they also miss features that are essential
in programming languages, such as input/output operations, exception handling, floating
point arithmetic, and side-effects. In general, these limitations are addressed in extraction
tools by adding some degree of informality to the system, such as realizations in Coq [12],
or semantic attachments in PVS [4]. Due to this informality, the extracted code may be
inconsistent with respect to the original specification.
With these limitations in mind, we have developed PVSio to extend the PVS formal
specification language with standard programming language features. Since PVSio is based
on semantic attachments, extracted code may not respect its original specification. However,
PVSio has been designed to minimize logical errors that may be introduced by semantic
attachments. First of all, PVSio is a conservative extension to the PVS prelude library.
There are no axioms in PVSio, therefore, FALSE cannot be proved with the help of PVSio
theories. Yet, expressions like RANDOM = RANDOM may evaluate to FALSE in the PVS ground
evaluator, while they are provable TRUE in the PVS theorem prover. However, since there is
a clean separation between the theorem prover and the ground evaluator, FALSE cannot be
proved from PVSio evaluations. Finally, PVSio has been designed to be minimally invasive.
That is, sources of problems such as side-effects are, by design, well-localized in PVSio
applications.
There are two logical weakness in PVSio applications: floating point arithmetic and input
operations. It is well-known that floating point arithmetic does not satisfy even simple real
arithmetic properties such as associativity. For that reason, functions in stdmath are not
identified with functions in the reals library. In other words, PVSio does not have axioms
such as SQ(SIN(x))+SQ(SIN(x))=1. On the negative side, there are no useful properties on
functions defined in stdmath. We expect to integrate a PVS formalization of floating point

arithmetic [3] with PVSio to allow reasoning about programs that use stdmath.
Side-effects produced by output operations are ignored at the logical level. For example,
print(s) is logically equivalent to TRUE in PVS. On the other hand, input operations are
problematic because their evaluations are not statically determined. This problem is classically solved by structuring the programs such that there is a phase for input and a phase
for processing data. The theory stdtokenizer is a good example of this method. Except
for the initializations of tokenizers that use input operations, all the other operations are
purely functional, i.e., they do not have side effects. Hence, we may formally reason about
a function like parse15 in PVS. Indeed, we have proved that parse15 terminates for any
tokenizer.
PVSio does not intend to be another programming language. It has been designed to serve
as a safe and efficient rapid prototyping tool for PVS. It lacks functionalities such as general
side-effects, i.e., assignments, low-level exception handling, and unbounded (and possibly
non-terminating) loops. These constructs are too invasive to be incorporated in a functional
specification language such as PVS. Functional counterparts of these constructs such as letin expressions, defensive programming, and recursive functions seem more appropriate for
this kind of specification language. Despite its limitations, PVSio is a powerful tool that
may handle non-trivial examples, such as a Conflict Detection and Resolution algorithm for
Air Traffic Management developed at the National Institute of Aerospace (formerly ICASE)
and NASA Langley [5]. In the future, new theories and functions are expected to be added.
We are particularly interested on supporting the interface of PVS with other environments
through PVSio.

[5] G. Dowek, C. Muñoz, and A. Geser. Tactical conflict detection and resolution in a 3-D
airspace. Technical Report NASA/CR-2001-210853 ICASE Report No. 2001-7, ICASENASA Langley, ICASE Mail Stop 132C, NASA Langley Research Center, Hampton VA
23681-2199, USA, April 2001.
[6] S. Owre, J. M. Rushby, and N. Shankar. PVS: A prototype verification system. In
Deepak Kapur, editor, 11th International Conference on Automated Deduction (CADE),
volume 607 of Lecture Notes in Artificial Intelligence, pages 748–752, Saratoga, NY,
June 1992. Springer-Verlag.
[7] Sam Owre and Natarajan Shankar. The formal semantics of PVS. Technical Report SRICSL-97-2, Computer Science Laboratory, SRI International, Menlo Park, CA, August
1997.
[8] N. Shankar. Efficiently executing PVS. Project report, Computer Science Laboratory,
SRI International, Menlo Park, CA, November 1999. Available at http://www.csl.
sri.com/shankar/PVSeval.ps.gz.
[9] G. L. Steele Jr. Common Lisp: The Language. Digital Press, Bedford, MA, second
edition, 1990.
[10] PVS Development Team. PVS 2.3 experimental features – ground evaluation. Available
at http://pvs.csl.sri.com/experimental/eval.html.
[11] PVS Development Team. PVS 3.1 release note. Available at http://pvs.csl.sri.
com/pvs-release-notes.html.
[12] The Coq Team. The Coq proof assistant: Reference manual: Version 7.2. Technical
Report RT-0255, INRIA, Rocquencourt, France, February 2002. Available at http:
//coq.inria.fr/doc/main.html.

21

Form Approved
OMB No. 0704-0188

REPORT DOCUMENTATION PAGE

The public reporting burden for this collection of information is estimated to average 1 hour per response, including the time for reviewing instructions, searching existing
data sources, gathering and maintaining the data needed, and completing and reviewing the collection of information. Send comments regarding this burden estimate or
any other aspect of this collection of information, including suggestions for reducing this burden, to Department of Defense, Washington Headquarters Services, Directorate
for Information Operations and Reports (0704-0188), 1215 Jefferson Davis Highway, Suite 1204, Arlington, VA 22202-4302. Respondents should be aware that
notwithstanding any other provision of law, no person shall be subject to any penalty for failing to comply with a collection of information if it does not display a currently
valid OMB control number.
PLEASE DO NOT RETURN YOUR FORM TO THE ABOVE ADDRESS.

1. REPORT DATE (DD-MM-YYYY)

01/12/2003

2. REPORT TYPE

3. DATES COVERED (From - To)

Contractor Report

4. TITLE AND SUBTITLE

5a. CONTRACT NUMBER

Rapid Prototyping in PVS
5b. GRANT NUMBER

NCC-1-02043
5c. PROGRAM ELEMENT NUMBER
5d. PROJECT NUMBER

6. AUTHOR(S)

Munoz, Cesar A.
5e. TASK NUMBER
5f. WORK UNIT NUMBER

786-10-10-10-00
7. PERFORMING ORGANIZATION NAME(S) AND ADDRESS(ES)

NASA Langley Research Center
Hampton, VA 23681-2199

National Institute of Aerospace (NIA)
144 Research Drive
Hampton, VA 23666

9. SPONSORING/MONITORING AGENCY NAME(S) AND ADDRESS(ES)

8. PERFORMING ORGANIZATION
REPORT NUMBER

NIA Report No. 2003-03

10. SPONSORING/MONITOR'S ACRONYM(S)

National Aeronautics and Space Administration
Washington, DC 20546-0001

NASA
11. SPONSORING/MONITORING
REPORT NUMBER

NASA/CR-2003-212418
12. DISTRIBUTION/AVAILABILITY STATEMENT

Unclassified - Unlimited
Subject Category 60, 61
Availability: NASA CASI (301) 621-0390

Distribution: Nonstandard

13. SUPPLEMENTARY NOTES

Langley Technical Monitor: Ricky Butler
14. ABSTRACT

PVSio is a conservative extension to the PVS prelude library that provides basic input/output capabilities to the PVS ground
evaluator. It supports rapid prototyping in PVS by enhancing the specification language with built-in constructs for string
manipulation, floating point arithmetic, and input/output operations.

15. SUBJECT TERMS

Rapid Prototyping; Functional Animation; Theorem Proving; PVS

16. SECURITY CLASSIFICATION OF:
a. REPORT

b. ABSTRACT

c. THIS PAGE

U

U

U

17. LIMITATION OF
ABSTRACT

UU

18. NUMBER 19b. NAME OF RESPONSIBLE PERSON
OF
STI Help Desk (email: help@sti.nasa.gov)
PAGES
19b. TELEPHONE NUMBER (Include area code)

26

(301) 621-0390
Standard Form 298 (Rev. 8-98)

Prescribed by ANSI Std. Z39-18

